
Overview

This document lists a TTCN change needed to correct issues in the ATS ‘iWD-EUTRA-B2011-03_D11wk49’.

Table of Contents

2Overview

Table of Contents
2
Corrections
2
Change 1 – GERAN preamble / race condition
2
Change 2 – GERAN component  / New global variable
4
Change 3 – GERAN preamble / New default handler
5
Change 4 – Correction to function f_TC_9_2_1_1_11_GERAN
5
Change 5 – Correction to f_TC_8_3_2_6_GERAN / …_UTRAN
8


Corrections 

Change 1 – GERAN preamble / race condition

Template name
f_GPRS_ReceiveUplinkMsg

Reason for change
TTCN is coded in such a way that first the TBF is established  and then the UL NAS message is received. But because different ports attached to  different stack layers in the SS  are used,  the order of arrival at the TTCN ports (G_RLC, G_CLLC and  G_LLC) may not be as expected. The current situation is unstable and race conditions arise.

For examble in the preamble of test case 6.2.2.6 the expected sequence is:

1. UE --> SS  PacketControlAcknowledgement (TLLI)  --> TTCN port G_RLC  (handled in function fl_ULTBFOnePhase )

2. TTCN -> SS stack  Assign.Reg(TLLI) on port G_CLLC

3. SS stack --> TTCN  Assign.Cnf on port G_CLLC

 (handled in function f_LLC_TLLI_Assign, called from ULTBFOnePhase)

4. UE --> SS Attach Request --> TTCN LLC_UnitData.Ind on port G_LLC

Because of the inherent asynchronous flows on different ports the order may not be as expected , e.g. the NAS message at step 4 may arrive before the confirmation at step3. 



Summary of change
The code has been extended with a special default handler for catching the uplink NAS mesage. This handler is activated just before the TBF is established.

TTCN module
GERAN_CommonFunctions.ttcn

MCC160 Comment


Before change

  function f_GPRS_ReceiveUplinkMsg (GERAN_CellId_Type p_CellId,

                                    template G_LLC_DATAMESSAGE_IND p_DataMsgInd,

                                    charstring p_StepNo := "") runs on GERAN_PTC  return G_LLC_DATAMESSAGE_IND

  {

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_ULDataInd;

    var integer v_RR_RA;

    var RFN v_RFN;

    G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase )) -> value v_ChanReq;

    v_RFN := v_ChanReq.access.rfn;

    v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

    if ((119 > v_RR_RA) or (v_RR_RA > 127)) {// establishment cause should be one phase access

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");

    }

    if (p_StepNo != "") {

      f_GERAN_PreliminaryPass (__FILE__, __LINE__, p_StepNo);

    }

    // finish configuring the TBF

    fl_ULTBFOnePhase (p_CellId, v_RFN, v_RR_RA);

    // Receive the message

    G_LLC.receive (p_DataMsgInd) -> value v_ULDataInd;

    // Now receive PCA that's sent when the TBF is released

    G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement));

    return v_ULDataInd; // return this to the calling function

  } // end of f_GPRS_ReceiveUplinkMsg 

After change

  function f_GPRS_ReceiveUplinkMsg (GERAN_CellId_Type p_CellId,

                                    template G_LLC_DATAMESSAGE_IND p_DataMsgInd,

                                    charstring p_StepNo := "") runs on GERAN_PTC  return G_LLC_DATAMESSAGE_IND

  {

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_ULDataInd;

    var integer v_RR_RA;

    var RFN v_RFN;

    G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase )) -> value v_ChanReq;

    v_RFN := v_ChanReq.access.rfn;

    v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

    if ((119 > v_RR_RA) or (v_RR_RA > 127)) {// establishment cause should be one phase access

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");

    }

    if (p_StepNo != "") {

      f_GERAN_PreliminaryPass (__FILE__, __LINE__, p_StepNo);

    }

    // finish configuring the TBF

    // Prepare for fetching UL NAS message and avoid race conditions

    vc_GERAN_Global.MsgInDefault.ulNAS := omit;

    var default v_MyDefaultVar := null;

    v_MyDefaultVar := activate(a_GERAN_ReceiveUlNasMsg ());

    fl_ULTBFOnePhase (p_CellId, v_RFN, v_RR_RA);

    // Receive the UL NAS message

    if(isvalue(vc_GERAN_Global.MsgInDefault.ulNAS)) {

       v_ULDataInd := vc_GERAN_Global.MsgInDefault.ulNAS;

    } else {

      G_LLC.receive (p_DataMsgInd) -> value v_ULDataInd;

    }

    // Reset global variable and default handler

    vc_GERAN_Global.MsgInDefault.ulNAS := omit;

    deactivate(v_MyDefaultVar);
    // Now receive PCA that's sent when the TBF is released

    G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement));

    return v_ULDataInd; // return this to the calling function

  } // end of f_GPRS_ReceiveUplinkMsg 

Change 2 – GERAN component  / New global variable

Function name
GERAN_MsgReceivedInDefault_Type

Reason for change
New global variable for catching UL messages.


Summary of change
New function global variable added to the GERAN component.

TTCN module
GERAN_CellInfo.ttcn

MCC160 Comment


Before change

  type record GERAN_MsgReceivedInDefault_Type {                       /* @status    APPROVED */

    EstablishIndType locationUpdatingRequest  optional,   

    LLCUnitDataIndType routingAreaUpdateRequest optional    

  };

After change

  type record GERAN_MsgReceivedInDefault_Type {                       

    EstablishIndType locationUpdatingRequest  optional,   

    LLCUnitDataIndType routingAreaUpdateRequest optional, 

    G_LLC_DATAMESSAGE_IND ulNAS optional // @RSfix

  };


Change 3 – GERAN preamble / New default handler 

Function name
a_GERAN_ReceiveUlNasMsg

Reason for change
New default handler for receiving UL NAS messages.



Summary of change
New function coded as shown below

TTCN module
GERAN_CommonFunctions.ttcn

MCC160 Comment


Before change



After change

  altstep a_GERAN_ReceiveUlNasMsg() runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_ulNas;

    //Receive any Routing Area Update Request message

    [] G_LLC.receive (car_G_LLC_UnitData_IND ( ?) ) -> value v_ulNas

    {

        vc_GERAN_Global.MsgInDefault.ulNAS  := v_ulNas;

    }

  }

Change 4 – Correction to function f_TC_9_2_1_1_11_GERAN

Function name
function f_TC_9_2_1_1_11_GERAN ()

Reason for change
Expected detach type is wrong in test step 21a8.

Summary of change
Amended the detach type.

TTCN module
NAS_Attach_UG_GERAN.ttcn

MCC160 Comment


Before change

  function f_TC_9_2_1_1_11_GERAN () runs on GERAN_PTC

  { /* Test Purpose (see f_TC_9_2_1_1_11_EUTRA) */

    /* @sic R5-113696: pc_GERAN and px_RATComb_Tested are checked at top-level sic@ */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test @sic R5s110318 sic@

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);

    v_RAC := f_GERAN_RAC_Get(geran_Cell24);

    v_LAC := f_GERAN_LAC_Get(geran_Cell24);

    f_GERAN_CreateCell(geran_Cell24);

    f_WaitForIRATCoOrd_Trigger (EUTRA); // @sic R5s110232 sic@

    // Pre-Preamble on GERAN cell first to set PTSMI, PTSMI Sig and TMSI in USIM

    f_GERAN_Pre_Preamble (geran_Cell24, PS_Only); // @sic R5s110176 sic@

    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_GERAN_TestBody_Set( true );

    //@siclog "Step 21a1" siclog@

    f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //  Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds? (F)

    //@siclog "Step 21a2" siclog@

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 30.0, "Step 21a2");

    //@siclog "Step 21a3" siclog@

    // The operator initiates an attach by MMI or by AT command.

    f_UT_TriggerAttach(UT);

    //  Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds? (F)

    //@siclog "Step 21a4" siclog@

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 30.0, "Step 21a4");

    //@siclog "Step 21a5" siclog@

    f_UT_SwitchOffUE (UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@

    //@siclog "Step 21a6" siclog@

    f_UT_SwitchOnUE (UT, false );

    //@siclog "Step 21a7 Void" siclog@

    //  Check: Does the UE transmit an ATTACH REQUEST message with IMSI-1and TMSI status indicated invalid TMSI?

    //@siclog "Step 21a8" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND(cr_G_AttachReqWithIMSI(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                                         f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                                         cr_RAI(omit),

                                                                                                         omit))); // @sic R5-113113 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 21a8");

    //@siclog "Step 22" siclog@

    // Send GPRS Attach Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24,

                           1,

                           cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                   tsc_LLCSapi_GMM,

                                                   tsc_LLC_PM,

                                                   px_GSM_CipheringOnOff,

                                                   cs_G_AttachAccept (cs_GMM_AttachResult(v_LLCDataInd.unit.msg.attReq.attachType.attachType),

                                                                      cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                      cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                      cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def))));

    //Receive GPRS Attach Complete Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND (cr_G_AttachComplete));

    f_GERAN_TestBody_Set( false );

    // Test case now finished on GERAN side

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_Postamble ( geran_Cell24, G1_IDLE, PS_Only); // @sic R5s110176 sic@

  } // f_TC_9_2_1_1_11_GERAN

  

After change

  function f_TC_9_2_1_1_11_GERAN () runs on GERAN_PTC

  { /* Test Purpose (see f_TC_9_2_1_1_11_EUTRA) */

    /* @sic R5-113696: pc_GERAN and px_RATComb_Tested are checked at top-level sic@ */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test @sic R5s110318 sic@

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);

    v_RAC := f_GERAN_RAC_Get(geran_Cell24);

    v_LAC := f_GERAN_LAC_Get(geran_Cell24);

    f_GERAN_CreateCell(geran_Cell24);

    f_WaitForIRATCoOrd_Trigger (EUTRA); // @sic R5s110232 sic@

    // Pre-Preamble on GERAN cell first to set PTSMI, PTSMI Sig and TMSI in USIM

    f_GERAN_Pre_Preamble (geran_Cell24, PS_Only); // @sic R5s110176 sic@

    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_GERAN_TestBody_Set( true );

    //@siclog "Step 21a1" siclog@

    f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //  Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds? (F)

    //@siclog "Step 21a2" siclog@

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 30.0, "Step 21a2");

    //@siclog "Step 21a3" siclog@

    // The operator initiates an attach by MMI or by AT command.

    f_UT_TriggerAttach(UT);

    //  Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds? (F)

    //@siclog "Step 21a4" siclog@

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 30.0, "Step 21a4");

    //@siclog "Step 21a5" siclog@

    f_UT_SwitchOffUE (UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@

    //@siclog "Step 21a6" siclog@

    f_UT_SwitchOnUE (UT, false );

    //@siclog "Step 21a7 Void" siclog@

    //  Check: Does the UE transmit an ATTACH REQUEST message with IMSI-1and TMSI status indicated invalid TMSI?

    //@siclog "Step 21a8" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND(cr_G_AttachReqWithIMSI(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined)),
                                                                                                         f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                                         cr_RAI(omit),

                                                                                                         omit))); // @sic R5-113113 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 21a8");

    //@siclog "Step 22" siclog@

    // Send GPRS Attach Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24,

                           1,

                           cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                   tsc_LLCSapi_GMM,

                                                   tsc_LLC_PM,

                                                   px_GSM_CipheringOnOff,

                                                   cs_G_AttachAccept (cs_GMM_AttachResult(v_LLCDataInd.unit.msg.attReq.attachType.attachType),

                                                                      cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                      cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                      cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def))));

    //Receive GPRS Attach Complete Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND (cr_G_AttachComplete));

    f_GERAN_TestBody_Set( false );

    // Test case now finished on GERAN side

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_Postamble ( geran_Cell24, G1_IDLE, PS_Only); // @sic R5s110176 sic@

  } // f_TC_9_2_1_1_11_GERAN

  

Change 5 – Correction to f_TC_8_3_2_6_GERAN / …_UTRAN

Function names
function f_TC_8_3_2_6_GERAN()
function f_TC_8_3_2_6_UTRAN()

Reason for change
As already mentioned in R5s110724, a deadlock situation is created during initialization by trying to synchronize all 3 RATs (EUTRA, UTRAN, GERAN) at the same time. 

Summary of change
Instead, synchronization should be done one by one (i.e. EUTRA with either GERAN or UTRAN).

TTCN modules
RRC_Measurement_GERAN
RRC_Measurement_UTRAN 

MCC160 Comment


Before change  –  example: function f_TC_8_3_2_6_GERAN(); 
similar change required for  function f_TC_8_3_2_6_UTRAN()
//================================================================

  /*

   * @desc      REFERENCE TS 36.523-1 clause 8.3.2.6

   * @status    APPROVED

   */

  function f_TC_8_3_2_6_GERAN() runs on GERAN_PTC

  { // Measurement configuration control and reporting / inter RAT measurements: Simultaneous A2 and two B2 (measurements of E-UTRAN, UTRAN and GERAN cells)

    var integer v_T0_RSSI_Cell24 := -85;

    var integer v_T1_RSSI_Cell24 := -65;

    //Initialise test case

    f_GERAN_Init(EUTRA_UTRAN_GERAN);
    f_GERAN_Send_IR_SysInfoToEUTRA(geran_Cell24);

    //Create GERAN cell and broadcast System Infomation

    f_GERAN_CreateCell_MinUL(geran_Cell24);

    //Trigger: Set cell power: T0

    f_WaitForIRATCoOrd_Trigger(EUTRA);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_PhyCh0, v_T0_RSSI_Cell24);

    f_GERAN_TestBody_Set(true);

    //------------------------------------------------

    //Start TestBody

    //@siclog "Step 4" siclog@

    //Trigger: Change cell power according to T1

    f_WaitForIRATCoOrd_Trigger(EUTRA);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_PhyCh0, v_T1_RSSI_Cell24);

    //Wait for the end of the test

    f_WaitForIRATCoOrd_Trigger(EUTRA);

    //End TestBody

    //------------------------------------------------

    f_GERAN_TestBody_Set(false);

    f_GERAN_SS_CellRelease(geran_Cell24, true); // @sic R5s110724 sic@

  }; //end of testcase 8.3.2.6



After change–  example: function f_TC_8_3_2_6_GERAN(); 
similar change required for  function f_TC_8_3_2_6_UTRAN()
  //================================================================

  /*

   * @desc      REFERENCE TS 36.523-1 clause 8.3.2.6

   * @status    APPROVED

   */

  function f_TC_8_3_2_6_GERAN() runs on GERAN_PTC

  { // Measurement configuration control and reporting / inter RAT measurements: Simultaneous A2 and two B2 (measurements of E-UTRAN, UTRAN and GERAN cells)

    var integer v_T0_RSSI_Cell24 := -85;

    var integer v_T1_RSSI_Cell24 := -65;

    //Initialise test case

    f_GERAN_Init(EUTRA_GERAN);
    f_GERAN_Send_IR_SysInfoToEUTRA(geran_Cell24);

    //Create GERAN cell and broadcast System Infomation

    f_GERAN_CreateCell_MinUL(geran_Cell24);

    //Trigger: Set cell power: T0

    f_WaitForIRATCoOrd_Trigger(EUTRA);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_PhyCh0, v_T0_RSSI_Cell24);

    f_GERAN_TestBody_Set(true);

    //------------------------------------------------

    //Start TestBody

    //@siclog "Step 4" siclog@

    //Trigger: Change cell power according to T1

    f_WaitForIRATCoOrd_Trigger(EUTRA);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_PhyCh0, v_T1_RSSI_Cell24);

    //Wait for the end of the test

    f_WaitForIRATCoOrd_Trigger(EUTRA);

    //End TestBody

    //------------------------------------------------

    f_GERAN_TestBody_Set(false);

    f_GERAN_SS_CellRelease(geran_Cell24, true); // @sic R5s110724 sic@

  }; //end of testcase 8.3.2.6

