
Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iWD-EUTRA-B2011-03_D11wk49’.

Corrections 

Change 1 – Correction to a_ICMPv6_Handler
Function name
a_ICMPv6_Handler

Reason for change
As specified in RFC 2460 clause 8.1, the parameter destination address used for the calculation of checksum in fl_ICMPv6_UpdateCRCChecksum should always be v_RemoteSocket.IpAddr.V6.Addr, that is, the destination address of the packet that is built like a response. In the function below, for the case of Multicast, the destination address used for that calculation is the future IPv6 address (vc_IP_PTC_PDN_AddrServer.ICMPv6[v_PdnIndex].ClientAddr) which is incorrect. 

Summary of change
The destination address parameter for the calculation of checksum in fl_ICMPv6_UpdateCRCChecksum was modified so that it is always set to v_RemoteSocket.IpAddr.V6.Addr.

TTCN module
IP_PTC_Main.ttcn        

MCC160 Comment


Before change

//----------------------------------------------------------------------------

  /*

   * @desc      handling of ICMPv6 messages sent by the UE

   * @status    APPROVED

   */

  altstep a_ICMPv6_Handler() runs on IP_PTC

  {

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var Datagram_UL_Type v_Datagram;

    var octetstring v_RespData;

    var charstring v_ServerIpAddr;

    var integer v_PdnIndex;

    var charstring v_ClientIpAddr;

    [] IP_SOCK.receive(cr_ICMPv6_DATA_IND(cr_IP_Socket(?, omit)))   /* allow any IP address to deal with multicasts */

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_Datagram := v_IP_SOCKET_IND.DATA.Ind.ICMP.RecvFrom;

        v_ServerIpAddr := v_LocalSocket.IpAddr.V6.Addr;         /* causes runtime error if address is V4; => SS implementation error */

        v_ClientIpAddr := v_RemoteSocket.IpAddr.V6.Addr;        /* causes runtime error if address is V4; => SS implementation error */

        if (fl_IPv6AddrIsMulticast(v_ServerIpAddr)) {           /* deal with multicast */

          /* NOTE:

             when the local link address (as assigned as interface identifier via NAS) for different PDNs is different

             it would also be possible to search for the server by looking for the client's local link address;

             but for simplification we use the DRB info as for DHCP */

          v_PdnIndex := fl_IP_AddrServer_Find(vc_IP_PTC_PDN_AddrServer.ICMPv6, v_Datagram.DrbInfo);

          v_ServerIpAddr := vc_IP_PTC_PDN_AddrServer.ICMPv6[v_PdnIndex].ServerAddr;

          v_ClientIpAddr := vc_IP_PTC_PDN_AddrServer.ICMPv6[v_PdnIndex].ClientAddr;

          v_LocalSocket.IpAddr.V6.Addr  := fl_IPv6AddrGetLinkLocalAddr(v_ServerIpAddr);  /* Assumption: response for a multicast needs to be sent with SRC address as local link address (as for ROUTER SOLICITATION)

                                                                                            @sic R5s110449 change 4 sic@ */

        }

        v_RespData := f_ICMPv6_GetResponse(v_Datagram.Buffer, v_ClientIpAddr);

        if (lengthof(v_RespData) >  0) {

          v_RespData := fl_ICMPv6_UpdateCRCChecksum(v_LocalSocket.IpAddr.V6.Addr, v_ClientIpAddr, v_RespData);

          IP_SOCK.send(cs_ICMPv6_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }

After change

//----------------------------------------------------------------------------

  /*

   * @desc      handling of ICMPv6 messages sent by the UE

   * @status    APPROVED

   */

  altstep a_ICMPv6_Handler() runs on IP_PTC

  {

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var Datagram_UL_Type v_Datagram;

    var octetstring v_RespData;

    var charstring v_ServerIpAddr;

    var integer v_PdnIndex;

    var charstring v_ClientIpAddr;

    [] IP_SOCK.receive(cr_ICMPv6_DATA_IND(cr_IP_Socket(?, omit)))   /* allow any IP address to deal with multicasts */

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_Datagram := v_IP_SOCKET_IND.DATA.Ind.ICMP.RecvFrom;

        v_ServerIpAddr := v_LocalSocket.IpAddr.V6.Addr;         /* causes runtime error if address is V4; => SS implementation error */

        v_ClientIpAddr := v_RemoteSocket.IpAddr.V6.Addr;        /* causes runtime error if address is V4; => SS implementation error */

        if (fl_IPv6AddrIsMulticast(v_ServerIpAddr)) {           /* deal with multicast */

          /* NOTE:

             when the local link address (as assigned as interface identifier via NAS) for different PDNs is different

             it would also be possible to search for the server by looking for the client's local link address;

             but for simplification we use the DRB info as for DHCP */

          v_PdnIndex := fl_IP_AddrServer_Find(vc_IP_PTC_PDN_AddrServer.ICMPv6, v_Datagram.DrbInfo);

          v_ServerIpAddr := vc_IP_PTC_PDN_AddrServer.ICMPv6[v_PdnIndex].ServerAddr;

          v_ClientIpAddr := vc_IP_PTC_PDN_AddrServer.ICMPv6[v_PdnIndex].ClientAddr;

          v_LocalSocket.IpAddr.V6.Addr  := fl_IPv6AddrGetLinkLocalAddr(v_ServerIpAddr);  /* Assumption: response for a multicast needs to be sent with SRC address as local link address (as for ROUTER SOLICITATION)

                                                                                            @sic R5s110449 change 4 sic@ */

        }

        v_RespData := f_ICMPv6_GetResponse(v_Datagram.Buffer, v_ClientIpAddr);

        if (lengthof(v_RespData) >  0) {

          v_RespData := fl_ICMPv6_UpdateCRCChecksum(v_LocalSocket.IpAddr.V6.Addr, v_RemoteSocket.IpAddr.V6.Addr, v_RespData);

          IP_SOCK.send(cs_ICMPv6_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }

Change 2 – Correction to cs_RouterAdvertisement

Function name
Template cs_RouterAdvertisement

Reason for change
As per RFC 4861, cl. 4.6.2, “Length” should be 4 instead of 3. 

Summary of change
optionLength set to 4 as per RFC 4861 cl. 4.6.2.

TTCN module
ICMPv6_Functions.ttcn

MCC160 Comment


Before change

template (value) ICMPv6Message cs_RouterAdvertisement(charstring p_ClientIpAddress) :=

  { /* @status    APPROVED */

    routerAdvMsg := {

      icmpType := tsc_ICMPv6_RtAdvMsg,  // RFC 4861

      icmpCode := 0,

      checksum := '0000'O,              // to be replaced before sending

      curHopLimit := 64,

      managedConfigFlag := '0'B,        // Not managed, i.e. stateless autoconfiguration

      otherConfigFlag := '0'B,          // FFS DHCPv6 not available

      reserved := '000000'B,

      routerLifetime := oct2int('FFFF'O), // largest posible value

      reachableTime := 0,               // unspecified

      retransTimer := 0,                // unspecified

      rtAdvOptions := {

        srcLinkLayerAddr := omit,

        mtuOption := omit,

        prefixInfoList := {

          {

            icmpType := tsc_ICMPv6_PrefixInfo,

            optionLength := 3,          // 24 = 3 * 8 octets is the length  for the complete option

            prefixLength := 64,         // 64 bits prefix

            linkFlag := '1'B,           // on-link flag set

            autoConfigFlag := '1'B,     // autonmous address-autconfiguration set

            reserved1 := '000000'B,

            validLifetime := oct2int ('FFFFFFFF'O),     // infinity

            preferredLifetime := oct2int('FFFFFFFF'O),  // infinity

            reserved2 := 0,

            // And here the 64 bits prefix

            prefix := substr (f_Convert_IPv6Addr2OctString(p_ClientIpAddress), 0, 8) & '0000000000000000'O

          }

        }, // prefixInfoList

        advertisementInterval := omit,

        homeAgentInfo := omit,

        otherOption := omit

      } // options

    }

  };

  

After change

template (value) ICMPv6Message cs_RouterAdvertisement(charstring p_ClientIpAddress) :=

  { /* @status    APPROVED */

    routerAdvMsg := {

      icmpType := tsc_ICMPv6_RtAdvMsg,  // RFC 4861

      icmpCode := 0,

      checksum := '0000'O,              // to be replaced before sending

      curHopLimit := 64,

      managedConfigFlag := '0'B,        // Not managed, i.e. stateless autoconfiguration

      otherConfigFlag := '0'B,          // FFS DHCPv6 not available

      reserved := '000000'B,

      routerLifetime := oct2int('FFFF'O), // largest posible value

      reachableTime := 0,               // unspecified

      retransTimer := 0,                // unspecified

      rtAdvOptions := {

        srcLinkLayerAddr := omit,

        mtuOption := omit,

        prefixInfoList := {

          {

            icmpType := tsc_ICMPv6_PrefixInfo,

            optionLength := 4,          // 32 = 4 * 8 octets is the length  for the complete option

            prefixLength := 64,         // 64 bits prefix

            linkFlag := '1'B,           // on-link flag set

            autoConfigFlag := '1'B,     // autonmous address-autconfiguration set

            reserved1 := '000000'B,

            validLifetime := oct2int ('FFFFFFFF'O),     // infinity

            preferredLifetime := oct2int('FFFFFFFF'O),  // infinity

            reserved2 := 0,

            // And here the 64 bits prefix

            prefix := substr (f_Convert_IPv6Addr2OctString(p_ClientIpAddress), 0, 8) & '0000000000000000'O

          }

        }, // prefixInfoList

        advertisementInterval := omit,

        homeAgentInfo := omit,

        otherOption := omit

      } // options

    }

  };

